
JAGS MODEL CODE 

Occupancy model  
 
sink("occ_.txt") 
cat("  
    model { 
     
    ############## 
    ### PRIORS ### 
    ############## 
    psi0 ~ dlogis(0,1) 
    beta_shdi ~ dlogis(0,1) 
    beta_fw ~ dlogis(0,1) 
    beta_road ~ dlogis(0,1) 
     
    p0   ~ dlogis(0,1) 
     
 
    ################## 
    ### LIKELIHOOD ### 
    ################## 
     
    # OCCUPANCY SUBMODEL - ECOLOGICAL PROCESS 
    for(i in 1:nsite){ 
    z[i] ~ dbern(psi[i]) 
     
    logit(psi[i]) <- psi0 + 
    beta_shdi * shdi[i] + 
    beta_road * road[i] + 
    beta_fw  * fw[i] 
     
    # DETECTION SUBMODEL - OBSERVATION PROCESS 
    for(j in 1:nsurvey){ 
    y[i,j] ~ dbern(z[i]*p[i,j]) 
     
    logit(p[i,j]) <- p0 
     
    ################### 
    # Goodness of fit # 
    ################### 
     
            # Replicated data 
            y_rep[i,j] ~ dbern(z[i]*p[i,j])  # replicated data 
             
            # Squared error for observed data 



            D_obs[i,j] <- (y[i,j] - z[i]*p[i,j])^2 
             
            # Squared error for replicated data 
            D_rep[i,j] <- (y_rep[i,j] - z[i]*p[i,j])^2 
     
    } # j survey 
     
    } # i site 
     
    #DERIVED PARAMETERS 
    sites.occ <- sum(z[])/74 
    p_overall <- mean(p[,]) 
 
    # Sum of squared errors 
    total_D_obs <- sum(D_obs[,]) 
    total_D_rep <- sum(D_rep[,]) 
 
    # Bayesian p-value 
    bp_value <- step(total_D_rep - total_D_obs) 
     
    } # end of model   
    ",fill = TRUE) 
sink() 
 
 
Royle-Nichols model  
 
sink("RN.txt") 
cat(" 
    model { 
    ############## 
    ### PRIORS ### 
    ############## 
    beta0 ~ dnorm(0,0.001) 
    beta_shdi ~ dnorm(0,0.001) 
    beta_fw ~ dnorm(0,0.001) 
    beta_road ~ dnorm(0,0.001) 
     
    p0 ~ dlogis(0,1) 
     
    ################## 
    ### LIKELIHOOD ### 
    ################## 
     
    for(i in 1:nsite){ 
        N[i] ~ dpois(lambda[i]) 



        log(lambda[i]) <- beta0 +  
                          beta_shdi * shdi[i] + 
                          beta_road * road[i] + 
                          beta_fw  * fw[i] 
     
        for (j in 1:nweek){ 
            y[i,j] ~ dbern(p[i,j])  # observed data 
            p[i,j] <- 1 - (1 - r[i,j])^(N[i]) 
            logit(r[i,j]) <- p0 
             
             
    ################### 
    # Goodness of fit # 
    ################### 
 
            # Generate replicated data 
            y_rep[i,j] ~ dbern(p[i,j])  # replicated data 
             
            # Squared error for observed data 
            D_obs[i,j] <- (y[i,j] - p[i,j])^2 
             
            # Squared error for replicated data 
            D_rep[i,j] <- (y_rep[i,j] - p[i,j])^2 
        } 
    } 
     
    ########################## 
    ### DERIVED PARAMETERS ### 
    ########################## 
     
    # Abundance: average across sites 
    abun <- sum(N[])/nsite 
     
    # Sum of squared errors 
    total_D_obs <- sum(D_obs[,]) 
    total_D_rep <- sum(D_rep[,]) 
 
    # Bayesian p-value 
    bp_value <- step(total_D_rep - total_D_obs) 
     
    } # end of model   
    ",fill = TRUE) 
sink() 
 
 
Poisson regression model 



 
sink("count.txt") 
cat("  
    model { 
     
    ############## 
    ### PRIORS ### 
    ############## 
     
    beta0 ~ dnorm(0, 0.001)   
    beta_shdi ~ dnorm(0, 0.001)   
    beta_fw ~ dnorm(0, 0.001)   
    beta_road ~ dnorm(0, 0.001)   
 
    tau <-  1/sigma^2 
    sigma ~ dunif(0,20) 
     
    ################## 
    ### LIKELIHOOD ### 
    ################## 
 
    for (i in 1:nsite) { 
    y[i] ~ dpois(lambda[i]) 
     
    log(lambda[i]) <- beta0 +  
    beta_shdi * shdi[i] + 
    beta_road * road[i] + 
    beta_fw  * fw[i]  + 
    log(effort[i]) + 
    eps[i] 
     
    eps[i] ~ dnorm(0,tau) 
     
    ################### 
    # Goodness of fit # 
    ################### 
     
    pr[i] <- (y[i] - lambda[i])/sqrt(lambda[i]) 
    pr2[i] <- pr[i]^2 
     
    y.new[i] ~ dpois(lambda[i]) 
     
    pr.new[i] <- (y.new[i] - lambda[i])/sqrt(lambda[i]) 
    pr.new2[i] <- pr.new[i]^2 
     
    } #i 



     
    #################### 
    #DERIVED PARAMETERS# 
    #################### 
     
    sites.count <- sum(lambda[])/74 
     
    pr2.sum <- sum(pr2[]) 
    pr.new2.sum <- sum(pr.new2[]) 
     
    # Bayesian p-value 
    bp <- pr2.sum > pr.new2.sum 
 
    } # end of model   
    ",fill = TRUE) 
sink() 
 
 
REST model 

sink("rho.txt") 
cat("model 
    { 
    for(i in 1:Nstay){ 
        is.censored[i] ~ dinterval(stay[i],cens[i]) 
        stay[i] ~ dlnorm(mean, tau) 
    } 
     
    mean ~ dgamma(0.01, 0.01) 
    tau ~ dgamma(0.01, 0.01)    
    tau2 <- 1/sigma^2 
    sigma ~ dunif(0,20) 
 
    for(i in 1:Ncam){ 
        y[i] ~ dpois(mu[i]) 
         
        eps[i] ~ dnorm(0,tau2) 
 
        log(mu[i]) <- log(S) + log(eff[i]) + log(rho[i]) + log(actv) - mean - 1/(2*tau) 
         
        log(rho[i]) <- 
            effects[1] + 
            effects[2] * shdi[i] + 
            effects[3] * fw[i] + 
            effects[4] * road[i] + 
            eps[i]                # extra site-level variance not captured by the covariates 



 
        # Calculate squared residuals for observed data 
        D_obs[i] <- ((y[i] - mu[i])/sqrt(mu[i]))^2  # Squared difference for observed data 
    } 
 
    for (v in 1:4) { 
        effects[v] ~ dnorm(0, 0.01) 
    } 
 
    # Calculate total squared error for observed data 
    total_D_obs <- sum(D_obs) 
 
    for (i in 1:Ncam) { 
        y_rep[i] ~ dpois(mu[i]) 
 
        # Calculate squared residuals for replicated data 
        D_rep[i] <- ((y_rep[i] - mu[i])/sqrt(mu[i]))^2  # Squared difference for replicated data 
    } 
 
    # Calculate total squared error for replicated data 
    total_D_rep <- sum(D_rep) 
 
    # Bayesian p-value 
    bp_value <- step(total_D_rep - total_D_obs) 
 
    D <- mean(rho[]) 
} 
") 
sink() 



Table S1. Model-estimated means, 95% credible intervals (95% CrI), Gelman-Rubin statistics, and Bayesian p-values of population-
level occupancy probability (occupancy), local abundance, relative abundance, and density (individuals/km2) for bobcat, coyote, 
fisher, gray fox, and red fox from camera trap images collected during summer 2021 in southeastern New Hampshire, USA. Results 
are based on three MCMC chains of 20,000 iterations each, a burn-in of 2,000 iterations, and a thinning rate of 1. 

Metric Species Posterior Mean Lower 95% CrI Upper 95% CrI Gelman-Rubin Bayesian p-value 
Occupancy Bobcat 0.65 0.42 0.95 1.00 0.52 

 Coyote 0.70 0.65 0.78 1.00 0.51 
 Fisher 0.38 0.24 0.59 1.00 0.53 
 Gray fox 0.11 0.07 0.32 1.00 0.55 
 Red fox 0.33 0.31 0.38 1.00 0.51 

Local abundance Bobcat 1.16 0.50 3.07 1.00 0.50 
 Coyote 1.76 1.16 2.95 1.00 0.66 
 Fisher 0.41 0.23 0.95 1.01 0.50 
 Gray fox 0.11 0.07 0.28 1.00 0.57 
 Red fox 0.47 0.36 0.64 1.00 0.76 

Relative abundance Bobcat 0.61 0.39 0.94 1.00 0.73 
 Coyote 2.45 2.10 2.82 1.00 0.57 
 Fisher 0.55 0.31 1.00 1.00 0.71 
 Gray fox 0.17 0.08 0.29 1.00 0.86 
 Red fox 2.14 1.82 2.48 1.00 0.62 

Density Bobcat 0.05 0.03 0.09 1.00 0.36 
 Coyote 0.22 0.16 0.28 1.00 0.43 
 Fisher 0.02 0.00 0.03 1.00 0.37 
 Gray fox NA NA NA NA NA 
 Red fox 0.28 0.19 0.38 1.02 0.45 

 

 



Table S2. Model-estimated detection probability and 95% credible intervals (95% CrI) for bobcat, coyote, fisher, gray fox, and red fox 
from camera trap images collected during summer 2021 in southeastern New Hampshire, USA. Results are based on three MCMC 
chains of 20,000 iterations each, a burn-in of 2,000 iterations, and a thinning rate of 1. 

Species Posterior Mean Lower 95% CrI Upper 95% CrI 
Bobcat 0.08 0.04 0.13 
Coyote 0.23 0.18 0.27 
Fisher 0.09 0.05 0.16 

Gray fox 0.16 0.04 0.31 
Red fox 0.26 0.20 0.32 

